In[1]:=
```
(*4-momenta in the LAB frame, before collision*)
```

In[2]:=
```
p1LAB = {m1 c / Sqrt[1 - v1^2 / c^2], m1 v1 / Sqrt[1 - v1^2 / c^2]}; MatrixForm[p1LAB]

p2LAB = {m2 c / Sqrt[1 - v2^2 / c^2], m2 v2 / Sqrt[1 - v2^2 / c^2]};

MatrixForm[p2LAB]
```

Out[2]//MatrixForm=

$$\begin{pmatrix} \frac{c\ m1}{\sqrt{1-\frac{v1^2}{c^2}}} \\ \frac{m1\ v1}{\sqrt{1-\frac{v1^2}{c^2}}} \end{pmatrix}$$

Out[3]//MatrixForm=

$$\begin{pmatrix} \frac{c\ m2}{\sqrt{1-\frac{v2^2}{c^2}}} \\ \frac{m2\ v2}{\sqrt{1-\frac{v2^2}{c^2}}} \end{pmatrix}$$

In[4]:=
```
(*Lorentz boost with general velocity v*)

Λ[v_] := 1 / Sqrt[1 - v^2 / c^2] {{1, -v / c}, {-v / c, 1}}

MatrixForm[Λ[v]]
```

Out[5]//MatrixForm=

$$\begin{pmatrix} \frac{1}{\sqrt{1-\frac{v^2}{c^2}}} & -\frac{v}{c\sqrt{1-\frac{v^2}{c^2}}} \\ -\frac{v}{c\sqrt{1-\frac{v^2}{c^2}}} & \frac{1}{\sqrt{1-\frac{v^2}{c^2}}} \end{pmatrix}$$

In[6]:=
```
(*Find the velocity V of the center of mass (COM) frame where total momentum is 0*)
Λ[v].(p1LAB + p2LAB) // MatrixForm
sol = Solve[(Λ[v].(p1LAB + p2LAB))[[2]] == 0, v] // FullSimplify
```

Out[6]//MatrixForm=

$$\begin{pmatrix} \dfrac{\frac{c\,m1}{\sqrt{1-\frac{v1^2}{c^2}}}+\frac{c\,m2}{\sqrt{1-\frac{v2^2}{c^2}}}}{\sqrt{1-\frac{v^2}{c^2}}} - \dfrac{v\left(\frac{m1\,v1}{\sqrt{1-\frac{v1^2}{c^2}}}+\frac{m2\,v2}{\sqrt{1-\frac{v2^2}{c^2}}}\right)}{c\sqrt{1-\frac{v^2}{c^2}}} \\ -\dfrac{v\left(\frac{c\,m1}{\sqrt{1-\frac{v1^2}{c^2}}}+\frac{c\,m2}{\sqrt{1-\frac{v2^2}{c^2}}}\right)}{c\sqrt{1-\frac{v^2}{c^2}}} + \dfrac{\frac{m1\,v1}{\sqrt{1-\frac{v1^2}{c^2}}}+\frac{m2\,v2}{\sqrt{1-\frac{v2^2}{c^2}}}}{\sqrt{1-\frac{v^2}{c^2}}} \end{pmatrix}$$

Solve: There may be values of the parameters for which some or all solutions are not valid.

Out[7]=
$$\left\{\left\{v \to \frac{m2\sqrt{1-\frac{v1^2}{c^2}}\,v2 + m1\,v1\sqrt{1-\frac{v2^2}{c^2}}}{m2\sqrt{1-\frac{v1^2}{c^2}} + m1\sqrt{1-\frac{v2^2}{c^2}}}\right\}\right\}$$

In[8]:=
```
V = (v /. sol)[[1]] (*V denotes the velocity of the COM frame in the LAB frame*)
```

Out[8]=
$$\frac{m2\sqrt{1-\frac{v1^2}{c^2}}\,v2 + m1\,v1\sqrt{1-\frac{v2^2}{c^2}}}{m2\sqrt{1-\frac{v1^2}{c^2}} + m1\sqrt{1-\frac{v2^2}{c^2}}}$$

In[9]:=
```
(*Transform to the COM frame, but don't write in v=V yet,
it would look very complicated*)
MatrixForm[p1COM = Λ[v].p1LAB]
MatrixForm[p2COM = Λ[v].p2LAB]
```

Out[9]//MatrixForm=

$$\begin{pmatrix} \dfrac{c\,m1}{\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v1^2}{c^2}}} - \dfrac{m1\,v\,v1}{c\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v1^2}{c^2}}} \\ -\dfrac{m1\,v}{\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v1^2}{c^2}}} + \dfrac{m1\,v1}{\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v1^2}{c^2}}} \end{pmatrix}$$

Out[10]//MatrixForm=

$$\begin{pmatrix} \dfrac{c\,m2}{\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}} - \dfrac{m2\,v\,v2}{c\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}} \\ -\dfrac{m2\,v}{\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}} + \dfrac{m2\,v2}{\sqrt{1-\frac{v^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}} \end{pmatrix}$$

In[11]:=
```
(*The important physical
 input: in the COM frame the collision flips the spatial part of the 4-
   momenta. This is the only way energy- and momentum-conservation can hold.*)
(*The 2nd p stands for prime in the variable names,
i.e. 4-momenta after collision: the 3-momentum is flipped, the energy is the same*)
MatrixForm[p1pCOM = {{1, 0}, {0, -1}}.p1COM]
MatrixForm[p2pCOM = {{1, 0}, {0, -1}}.p2COM]
```

Out[11]//MatrixForm=

$$\begin{pmatrix} \frac{c\,m1}{\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v1^2}{c^2}}} - \frac{m1\,v\,v1}{c\,\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v1^2}{c^2}}} \\ \frac{m1\,v}{\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v1^2}{c^2}}} - \frac{m1\,v1}{\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v1^2}{c^2}}} \end{pmatrix}$$

Out[12]//MatrixForm=

$$\begin{pmatrix} \frac{c\,m2}{\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v2^2}{c^2}}} - \frac{m2\,v\,v2}{c\,\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v2^2}{c^2}}} \\ \frac{m2\,v}{\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v2^2}{c^2}}} - \frac{m2\,v2}{\sqrt{1-\frac{v^2}{c^2}}\,\sqrt{1-\frac{v2^2}{c^2}}} \end{pmatrix}$$

In[13]:=
```
(*Transform back to the LAB frame: Lorentz transformation with -v. *)
FullSimplify[MatrixForm[p1pLAB = Λ[-v].p1pCOM]]
FullSimplify[MatrixForm[p2pLAB = Λ[-v].p2pCOM]]
```

Out[13]//MatrixForm=

$$\begin{pmatrix} \frac{c\,m1\,\left(c^2+v\,(v-2\,v1)\right)}{\left(c^2-v^2\right)\,\sqrt{1-\frac{v1^2}{c^2}}} \\ -\frac{m1\,\left(v^2\,v1+c^2\,(-2\,v+v1)\right)}{\left(c^2-v^2\right)\,\sqrt{1-\frac{v1^2}{c^2}}} \end{pmatrix}$$

Out[14]//MatrixForm=

$$\begin{pmatrix} \frac{c\,m2\,\left(c^2+v\,(v-2\,v2)\right)}{\left(c^2-v^2\right)\,\sqrt{1-\frac{v2^2}{c^2}}} \\ -\frac{m2\,\left(v^2\,v2+c^2\,(-2\,v+v2)\right)}{\left(c^2-v^2\right)\,\sqrt{1-\frac{v2^2}{c^2}}} \end{pmatrix}$$

In[15]:=
```
(*We get the end result,
i.e. the 4-momenta expressions after collision in the LAB frame,
in terms of the masses and incoming velocities, if we substitute V in the place of v*)
FullSimplify[MatrixForm[p1pLAB /. v → V]]
FullSimplify[MatrixForm[p2pLAB /. v → V]]
```

Out[15]//MatrixForm=

$$\begin{pmatrix} -\dfrac{c\,m1\left(v2\left(-m1^2\,v2+m2^2\,(-2\,v1+v2)\right)+c^2\left(m1^2+m2^2+2\,m1\,m2\sqrt{1-\frac{v1^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}\right)\right)}{m1^2\sqrt{1-\frac{v1^2}{c^2}}\,v2^2+m2^2\sqrt{1-\frac{v1^2}{c^2}}\,v2^2+2\,m1\,m2\,v1\,v2\sqrt{1-\frac{v2^2}{c^2}}-c^2\left(m1^2\sqrt{1-\frac{v1^2}{c^2}}+m2^2\sqrt{1-\frac{v1^2}{c^2}}+2\,m1\,m2\sqrt{1-\frac{v1^2}{c^2}}\right)} \\ -\dfrac{m1\left(-\left(\left(m1^2+m2^2\right)v1\,v2^2\right)+c^2\left(m1^2\,v1-m2^2\,(v1-2\,v2)+2\,m1\,m2\sqrt{1-\frac{v1^2}{c^2}}\,v2\sqrt{1-\frac{v2^2}{c^2}}\right)\right)}{m1^2\sqrt{1-\frac{v1^2}{c^2}}\,v2^2+m2^2\sqrt{1-\frac{v1^2}{c^2}}\,v2^2+2\,m1\,m2\,v1\,v2\sqrt{1-\frac{v2^2}{c^2}}-c^2\left(m1^2\sqrt{1-\frac{v1^2}{c^2}}+m2^2\sqrt{1-\frac{v1^2}{c^2}}+2\,m1\,m2\sqrt{1-\frac{v2^2}{c^2}}\right)} \end{pmatrix}$$

Out[16]//MatrixForm=

$$\begin{pmatrix} -\dfrac{c\,m2\left(v1\left(-m2^2\,v1+m1^2\,(v1-2\,v2)\right)+c^2\left(m1^2+m2^2+2\,m1\,m2\sqrt{1-\frac{v1^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}\right)\right)}{2\,m1\,m2\,v1\sqrt{1-\frac{v1^2}{c^2}}\,v2+m1^2\,v1^2\sqrt{1-\frac{v2^2}{c^2}}+m2^2\,v1^2\sqrt{1-\frac{v2^2}{c^2}}-c^2\left(2\,m1\,m2\sqrt{1-\frac{v1^2}{c^2}}+m1^2\sqrt{1-\frac{v1^2}{c^2}}+m2^2\sqrt{1-\frac{v2^2}{c^2}}\right)} \\ -\dfrac{m2\left(-\left(\left(m1^2+m2^2\right)v1^2\,v2\right)+c^2\left(m1^2\,(2\,v1-v2)+m2^2\,v2+2\,m1\,m2\,v1\sqrt{1-\frac{v1^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}\right)\right)}{2\,m1\,m2\,v1\sqrt{1-\frac{v1^2}{c^2}}\,v2+m1^2\,v1^2\sqrt{1-\frac{v2^2}{c^2}}+m2^2\,v1^2\sqrt{1-\frac{v2^2}{c^2}}-c^2\left(2\,m1\,m2\sqrt{1-\frac{v1^2}{c^2}}+m1^2\sqrt{1-\frac{v1^2}{c^2}}+m2^2\sqrt{1-\frac{v2^2}{c^2}}\right)} \end{pmatrix}$$

In[17]:=
```
(*Checking 4-momentum conservation*)
p1LAB + p2LAB // MatrixForm
p1pLAB + p2pLAB // MatrixForm // FullSimplify
(*The two should be equal, if the transformation velocity v is V,
which brings us into the COM frame*)
p1LAB + p2LAB == p1pLAB + p2pLAB /. v → V // FullSimplify
```

Out[17]//MatrixForm=

$$\begin{pmatrix} \dfrac{c\,m1}{\sqrt{1-\frac{v1^2}{c^2}}}+\dfrac{c\,m2}{\sqrt{1-\frac{v2^2}{c^2}}} \\ \dfrac{m1\,v1}{\sqrt{1-\frac{v1^2}{c^2}}}+\dfrac{m2\,v2}{\sqrt{1-\frac{v2^2}{c^2}}} \end{pmatrix}$$

Out[18]//MatrixForm=

$$\begin{pmatrix} \dfrac{c\,m2\,v\sqrt{1-\frac{v1^2}{c^2}}\,(v-2\,v2)+c\,m1\,v\,(v-2\,v1)\sqrt{1-\frac{v2^2}{c^2}}+c^3\left(m2\sqrt{1-\frac{v1^2}{c^2}}+m1\sqrt{1-\frac{v2^2}{c^2}}\right)}{\left(c^2-v^2\right)\sqrt{1-\frac{v1^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}} \\ \dfrac{-m2\,v^2\sqrt{1-\frac{v1^2}{c^2}}\,v2-m1\,v^2\,v1\sqrt{1-\frac{v2^2}{c^2}}+c^2\left(m2\sqrt{1-\frac{v1^2}{c^2}}\,(2\,v-v2)+m1\,(2\,v-v1)\sqrt{1-\frac{v2^2}{c^2}}\right)}{\left(c^2-v^2\right)\sqrt{1-\frac{v1^2}{c^2}}\sqrt{1-\frac{v2^2}{c^2}}} \end{pmatrix}$$

Out[19]= True